# -------------------------------------------------
# Project for StorageUsage
# -------------------------------------------------

DESTDIR = output
TEMPLATE = app
CONFIG -= app_bundle
CONFIG += debug_and_release

CONFIG(debug, debug|release) {
    TARGET = storageusage-dbg
} else {
    TARGET = storageusage
}

DEPENDPATH = sources
INCLUDEPATH += sources

# Temp folders
MOC_DIR = temp/moc
OBJECTS_DIR = temp/obj
RCC_DIR = temp/rcc
UI_DIR = temp/ui

# Project files
SOURCES +=      main.cpp \
                mainwindow.cpp \
                scanwindow.cpp \
                usagegrid.cpp \
                scanthread.cpp \
                volumeview.cpp \
                sortfilter.cpp \
                delegate.cpp

HEADERS +=      mainwindow.h \
                scanwindow.h \
                usagegrid.h \
                scanthread.h \
                volumeview.h \
                sortfilter.h \
                delegate.h

RESOURCES += resources/storageusage.qrc

OTHER_FILES +=  debian/* \
                sources/storageusage.desktop

TRANSLATIONS += translations/*.ts

CONFIG(release, debug|release) {
    QMAKE_POST_LINK += strip $$DESTDIR/storageusage;
}

# profile
#QMAKE_CXXFLAGS += -pg
#LIBS += -pg
# ~profile

include(storageusage.pri)
